
• Το παρόν εκπαιδευτικό υλικό υπόκειται σε
άδειες χρήσης Creative Commons. 

• Για εκπαιδευτικό υλικό, όπως εικόνες, που 
υπόκειται σε άλλου τύπου άδειας χρήσης, η 
άδεια χρήσης αναφέρεται ρητώς. 

• Το παρόν εκπαιδευτικό υλικό έχει αναπτυχθεί στα πλαίσια 
του εκπαιδευτικού έργου του διδάσκοντα.

• Το έργο «Ανοικτά Ακαδημαϊκά Μαθήματα στο 
Αριστοτέλειο Πανεπιστήμιο Θεσσαλονίκης» έχει 
χρηματοδοτήσει μόνο τη αναδιαμόρφωση του 
εκπαιδευτικού υλικού. 

• Το έργο υλοποιείται στο πλαίσιο του Επιχειρησιακού 
Προγράμματος «Εκπαίδευση και Δια Βίου Μάθηση» και 
συγχρηματοδοτείται από την Ευρωπαϊκή Ένωση 
(Ευρωπαϊκό Κοινωνικό Ταμείο) και από εθνικούς πόρους.

• Υπάρχουν 3 κιβώτια a, b, c, διαφορετικού 
μεγέθους 10, 20, 30, πάνω σε ένα τραπέζι

• Στόχος η μετακίνηση και το στοίβαγμα των 
κιβωτίων το ένα πάνω στο άλλο, ώστε το 
μεγαλύτερο να είναι κάτω

• Ο βραχίονας που μετακινεί τα κιβώτια μπορεί 
να «σηκώσει» μόνο ένα κιβώτιο τη φορά

Περιγραφή Προβλήματος

Αρχική και Τελική Κατάσταση

(deftemplate block
(slot name)
(slot size (type INTEGER) 

(range 1 ?VARIABLE))
(slot place (type SYMBOL) 

(allowed-symbols 
heap hand stack)

(default ?DERIVE)))

(deffacts blocks
(block (name a) (size 10))
(block (name b) (size 20))
(block (name c) (size 30)))

Πρότυπο & Γεγονότα block

7



Αριστοτέλειο
Πανεπιστήμιο
Θεσσαλονίκης

Συστήματα Γνώσης

Τμήμα Πληροφορικής

(deftemplate goal

(slot name 

(type SYMBOL) 

(allowed-symbols stack 

put_down) 

(default ?DERIVE))

(slot parameter 

(type INTEGER) 

(range 1 ?VARIABLE))

)

Πρότυπο Στόχου (goal)

(defrule begin
(initial-fact)
=>
(assert (goal (name stack) 

(parameter 1))))

(defrule stop
?x <- (goal (name stack))
(not (block (place heap)))
=>
(retract ?x)
(halt))

Αρχικός & Τελικός Κανόνας

(defrule pickup1

(goal (name stack) (parameter ?N))

?x <- (block (name ?B1) (size ?S1) 

(place 

heap))

(not 

(and 

(block (name ?B2&~?B1) (size ?S2)

(place heap))

(test (> ?S2 ?S1))))

=>

(modify ?x (place hand)))

Κανόνας ανασηκώματος κιβωτίου

(defrule pickup1

(goal (name stack) (parameter ?N))

?x <- (block (name ?B1) (size ?S1) 

(place 

heap))

(not (block (name ?B2&~?B1) 

(size ?S2&:(> ?S2 ?S1))

(place heap)

)

)

=>

(modify ?x (place hand)))

Κανόνας ανασηκώματος κιβωτίου –
Συντομευμένη σύνταξη

(defrule holding

?x <- (goal (name stack))

(block (place hand))

=>

(modify ?x (name put_down))

)

Κανόνας μεταφοράς κιβωτίου

(defrule put_down

?x <- (goal (name put_down) 

(parameter ?N))

?y <- (block (place hand))

=>

(modify ?y (place stack))

(modify ?x (name stack) 

(parameter (+ ?N 1))))

Κανόνας τοποθέτησης κιβωτίου

CLIPS> (reset)

CLIPS> (facts)

f-0     (initial-fact)

f-1     (block (name a) (size 10) (place heap))

f-2     (block (name b) (size 20) (place heap))

f-3     (block (name c) (size 30) (place heap))

For a total of 4 facts.

CLIPS> (run)

CLIPS> (facts)

f-0     (initial-fact)

f-7     (block (name c) (size 30) (place stack))

f-11    (block (name b) (size 20) (place stack))

f-15    (block (name a) (size 10) (place stack))

For a total of 4 facts.

Εκτέλεση Προγράμματος

• Κάθε block πηγαίνει από το heap κατευθείαν στο 
stack, χωρίς να μεσολαβήσει το hand

• Πρότυπο & Γεγονότα block παραμένουν σχεδόν τα 
ίδια
– Λείπει το hand

• Δεν χρειάζονται τα γεγονότα goal

(deftemplate block
(slot name)
(slot size (type INTEGER)

(range 1 ?VARIABLE))
(slot place (type SYMBOL) 

(allowed-symbols heap stack)
(default ?DERIVE)))

Συντομότερη Λύση

(defrule begin

(initial-fact)

=>

(assert (stack 1)))

(defrule stop

?x <- (stack ?N)

(not (block (place heap)))

=>

(retract ?x)

(halt))

Αρχικός & Τελικός Κανόνας

(defrule move_to_stack

?x <- (stack ?N)

?y <- (block (name ?B1) (size ?S1)

(place heap))

(not (and (block (name ?B2&~?B1) 

(size ?S2) (place heap))

(test (> ?S2 ?S1))))

=>

(modify ?y (place stack))

(retract ?x)

(assert (stack (+ ?N 1))))

Κανόνας Μετακίνησης Κιβωτίου

CLIPS> (reset)

CLIPS> (facts)

f-0     (initial-fact)

f-1     (block (name a) (size 10) (place heap))

f-2     (block (name b) (size 20) (place heap))

f-3     (block (name c) (size 30) (place heap))

For a total of 4 facts.

CLIPS> (run)

CLIPS> (facts)

f-0     (initial-fact)

f-7     (block (name c) (size 30) (place stack))

f-11    (block (name b) (size 20) (place c))

f-15    (block (name a) (size 10) (place b))

For a total of 4 facts.

Εναλλακτική Αναπαράσταση Λύσης

(deftemplate block
(slot name)
(slot size (type INTEGER) 

(range 1 ?VARIABLE))
(slot place))

(deffacts blocks
(block (name a) (size 10)

(place heap))
(block (name b) (size 20)

(place heap))
(block (name c) (size 30)

(place heap)))

Νέο Πρότυπο & Γεγονότα block

(defrule put_down1

?x <- (goal (name put_down) 

(parameter ?N))

?y <- (block (place hand))

(not (block (place stack)))

=>

(modify ?y (place stack))

(modify ?x (name stack) 

(parameter (+ ?N 1))))

Νέος Κανόνας τοποθέτησης κιβωτίου (1)

(defrule put_down1

?x <- (goal (name put_down) 

(parameter 1))

?y <- (block (place hand))

=>

(modify ?y (place stack))

(modify ?x (name stack) 

(parameter 2)))

Νέος Κανόνας τοποθέτησης κιβωτίου (1) -
εναλλακτικός

(defrule put_down2

?x <- (goal (name put_down) 

(parameter ?N))

?y <- (block (place hand))

(block (name ?B) 

(place ~heap&~hand))

(not (block (place ?B)))

=>

(modify ?y (place ?B))

(modify ?x (name stack) 

(parameter (+ ?N 1))))

Νέος Κανόνας τοποθέτησης κιβωτίου (2)

